Yves Debard

Institut Universitaire de Technologie du Mans
Département Génie Mécanique et Productique

http://iut.univ-lemans.fr/ydlogi/index.html

24 mars 2006 – 29 mars 2011

Résistance des matériaux

Énergie de déformation : théorème de Castigliano (exercices avec Maple)

1 Poutre soumise à un effort normal

1.1 Énoncé

La poutre $(1-2)$ de longueur L, de section droite constante est encastrée en 1. Soient A l'aire de la section droite et E le module de Young du matériau.



La poutre porte en 2 une force de composantes $(F,\,0,\,0)$ et sur toute sa longueur, une force uniformément répartie d'intensité linéique p.

1. Calculer le déplacement du nœud 2.
2. Calculer le déplacement de la section d'abscisse x.

1.2 Solution

1. L'énergie de déformation est égale à :

$$E_{def} = \int_0^L \frac{N^2}{2\,EA}\,dx \quad \text{avec} \quad N(x) = F + p\,(L-x)$$

   Le déplacement horizontal du point 2 est donné par le théorème de Castigliano :

$$u_2 = \frac{\partial E_{def}}{\partial F}$$

2. Introduisons une force auxiliaire Q dans la section d'abscisse x :

$$E_{def} = \int_0^L \frac{N^2}{2\,EA}\,ds \quad \text{avec} \quad \begin{cases} N(s) = F + p\,(L-s) + Q & \text{si } 0 \le s < x \\ N(s) = F + p\,(L-s) & \text{si } x < s \le L \end{cases}$$

$$u(x) = \left.\frac{\partial E_{def}(Q)}{\partial Q}\right|_{Q=0}$$



Programme Maple :

```
# étude d'une poutre console soumise à un effort normal

restart:

# effort normal dans la section d'abscisse x

N:=F+p*(L-x);

# énergie de déformation

dEdef:=N^2/(2*E*A);
Edef:=int(dEdef,x=0..L);

# déplacement du noeud 2

u2:=simplify(diff(Edef,F));

# déplacement de la section d'abscisse x

N:=F+p*(L-s)+Q; # effort normal entre 0 et x
Edef:=int(N^2,s=0..x):
Edef:=simplify(Edef/(2*E*A));
ux:=diff(Edef,Q):
ux:=simplify(subs(Q=0,ux));
```

2 Poutre console sollicitée en flexion simple : influence de l'effort tranchant

2.1 Énoncé

La poutre $(1-2)$ de longueur L et de section droite constante est encastrée en 1.



Soient A , I_z et k_y les caractéristiques de la section droite.

E et ν sont respectivement le module de Young et le coefficient de Poisson du matériau.

Elle porte en 2 une force de composantes $(0, F, 0)$.

1. Calculer la flèche en 2.
2. La section droite est rectangulaire :

$$A = b\,h \quad , \quad I_z = \frac{b\,h^3}{12} \quad , \quad k_y = \frac{10\,(1+\nu)}{12+11\,\nu}$$

   Étudier l'influence de l'effort tranchant en fonction de $\dfrac{L}{h}$ pour plusieurs valeurs du coefficient de Poisson.

2.2 Solution

L'énergie de déformation est égale à :

$$E_{def} = \int_0^L dE_{def} = E_{def}(T_y) + E_{def}(Mf_z)$$

avec :

$$E_{def}(T_y) = \int_0^L \frac{T_y^2}{2\,GAk_y}\,dx \quad , \quad T_y = F$$

$$E_{def}(Mf_z) = \int_0^L \frac{Mf_z^2}{2\,EI_z}\,dx \quad , \quad Mf_z = F\,(L-x)$$

Le déplacement suivant y du nœud 2 est donné par le théorème de Castigliano :

$$v_2 = \frac{\partial E_{def}}{\partial F} = v_2(T_y) + v_2(Mf_z)$$

Programme Maple :

```
# étude d'une poutre console : influence de l'effort tranchant

restart:

# module d'élasticité transversal

G:=E/2/(1+nu);

# efforts dans la section d'abscisse x

Ty:=F;
Mfz:=F*(L-x);

# énergie de déformation

ETy:=int(Ty^2/(2*G*A*ky),x=0..L);
EMfz:=int(Mfz^2/(2*E*Iz),x=0..L);

# déplacement vertical du noeud 2

v2_Ty:=diff(ETy,F);v2_Mfz:=diff(EMfz,F);
v2:=v2_Ty+v2_Mfz;

# la section droite est rectangulaire

A:=b*h;Iz:=b*h^3/12;
ky:=10*(1+nu)/(12+11*nu);
plot(ky,nu=0..0.5,title="ky=10*(1+nu)/(12+11*nu)");

# influence de l'effort tranchant

L:=x*h;
f:=simplify(v2_Ty/v2);
f1:=subs(nu=0,f):
f2:=subs(nu=0.25,f):
f3:=subs(nu=0.5,f):
plot([f1*100,f2*100,f3*100],x=3..20,
title="Influence en % de l'effort tranchant en fonction du coefficient de Poisson",
labels=["L/h",""],color=[black,red,green],
legend=["nu=0","nu=0.25","nu=0.5"],thickness=2);
```

3 Poutre console sollicitée en flexion simple : matrice de rigidité

3.1 Énoncé

Soit la poutre $(1-2)$ de longueur L et de section droite constante. Soient A et I_z les caractéristiques de la section droite. E et ν sont respectivement le module de Young et le coefficient de Poisson du matériau.



La poutre est encastrée en 1.

Elle porte en 2 une force de composantes $(0, F, 0)$ et un couple de composantes $(0, 0, M)$.

1. Calculer les déplacements du nœud 2.
2. En déduire l'expression de la matrice de souplesse et de la matrice de rigidité de la structure.

3.2 Solution

Les efforts dans la section d'abscisse x sont :

$$T_y(x) = F$$

$$Mf_z(x) = M + F\,(L - x)$$

Les déplacements du nœud 2 sont donnés par le théorème de Castigliano :

$$v_2 = \frac{\partial E_{def}}{\partial F} \quad , \quad \theta_{z2} = \frac{\partial E_{def}}{\partial M}$$

avec

$$E_{def} = \int_0^L dE_{def} \quad , \quad dE_{def} = \frac{T_y^2}{2\,GA\,k_y}\,dx + \frac{Mf_z^2}{2\,EI_z}\,dx$$

On en déduit :

$$\begin{Bmatrix} v_2 \\ \theta_{z2} \end{Bmatrix} = [C] \begin{Bmatrix} F \\ M \end{Bmatrix}$$

où $[C]$ est la matrice de souplesse,

$$\begin{Bmatrix} F \\ M \end{Bmatrix} = [K] \begin{Bmatrix} v_2 \\ \theta_{z2} \end{Bmatrix}$$

où $[K] = [C]^{-1}$ est la matrice de rigidité.

Programme Maple :

```
# étude d'une poutre console

restart:with(linalg): # initialisation

# efforts dans la section d'abscisse x

Ty:=F;
Mfz:=M+F*(L-x);

# énergie de déformation

dEdef:=Ty^2/(2*G*A*ky)+Mfz^2/(2*E*Iz);
Edef:=int(dEdef,x=0..L);

# déplacements du noeud 2

u2:=grad(Edef,[F,M]);simplify(u2);

# matrice de souplesse

C:=hessian(Edef,[F,M]);
# ou
C:=jacobian(u2,[F,M]);

# matrice de rigidité

K:=inverse(C);
```

4 Poutre console sollicitée en flexion simple

4.1 Énoncé

Soit la poutre $(1-2)$ de longueur L et de section droite constante. Soient A et I_z les caractéristiques de la section droite. E et ν sont respectivement le module de Young et le coefficient de Poisson du matériau.



La poutre est encastrée en 1.

Elle porte sur toute sa longueur une force uniformément répartie dont l'intensité linéique est $(0, p, 0)$.

1. Calculer les déplacements du nœud 2.
2. La section droite est rectangulaire :

$$A = b\,h \quad , \quad I_z = \frac{b\,h^3}{12} \quad , \quad k_y = \frac{10\,(1+\nu)}{12+11\,\nu}$$

   Étudier l'influence de l'effort tranchant en fonction de $\dfrac{L}{h}$ pour plusieurs valeurs du coefficient de Poisson.

4.2 Solution

Introduisons en 2 une force et un couple auxiliaires F et M.

Les efforts dans la section d'abscisse x sont :

$$T_y(x) = F + p\,(L-x)$$

$$Mf_z(x) = M + F\,(L-x) + p\,\frac{(L-x)^2}{2}$$

Les déplacements du nœud 2 sont donnés par le théorème de Castigliano :

$$v_2 = \left.\frac{\partial E_{def}}{\partial F}\right|_{F=0\,,\,M=0} \quad , \quad \theta_{z2} = \left.\frac{\partial E_{def}}{\partial M}\right|_{F=0\,,\,M=0}$$

avec

$$E_{def} = \int_0^L dE_{def} \quad , \quad dE_{def} = \frac{T_y^2}{2\,GAk_y}\,dx + \frac{Mf_z^2}{2\,EI_z}\,dx$$

Programme Maple :

```
# étude d'une poutre console

restart: # initialisation

# efforts dans la section d'abscisse x

Ty:=F+q*(L-x);
Mfz:=M+F*(L-x)+q*(L-x)^2/2;

# énergie de déformation

ET:=int(Ty^2/(2*G*A*ky),x=0..L);
EMfz:=int(Mfz^2/(2*E*Iz),x=0..L);
Edef:=ET+EMfz;

# déplacements du noeud 2

v2T:=diff(ET,F);v2:=diff(Edef,F);
rotz2T:=diff(ET,M);rotz2:=diff(Edef,M);
F:=0;M:=0; v2;rotz2;

# influence de l'effort tranchant

G:=E/2/(1+nu);
ky:=10*(1+nu)/(12+11*nu);
A:=b*h:Iz:=b*h^3/12:

L:=x*h:
f:=simplify(v2T/v2);

f1:=subs(nu=0,f);f2:=subs(nu=0.25,f);f3:=subs(nu=0.5,f);
plot([f1*100,f2*100,f3*100],x=3..20,
legend=["nu=0","nu=0.25","nu=0.5"],
title="Influence de l'effort tranchant en % en fonction du coefficient de Poisson",
labels=["L / h","v2T/v2 en %"],
color=[black,red,green],thickness=2);
```

5 Étude d'un arc plan

5.1 Énoncé

L'arc $(1-2)$ de centre O et de rayon moyen R est encastré en 1. Il a une section droite constante. Soient A , I_z et k les caractéristiques de la section droite. E et ν sont respectivement le module de Young et le coefficient de Poisson du matériau.



L'arc est soumis en 2 à une force horizontale de composantes $(F_x, F_y, 0)$ et à un couple $(0, 0, M_z)$.

Calculer les déplacements du nœud 2.

5.2 Solution

Les efforts dans la section φ sont :

$$N(\varphi) = F_x \sin\varphi - F_y \cos\varphi$$

$$T(\varphi) = F_x \cos\varphi + F_y \sin\varphi$$

$$Mf_z(\varphi) = F_x R \sin\varphi + F_y R (1-\cos\varphi) + M_z$$

L'énergie de déformation est égale à :

$$E_{def} = \int_0^{3\pi/2} dE_{def} \quad \text{avec} \quad dE_{def} = \left(\frac{N^2}{2\,EA} + \frac{T^2}{2\,GA\,k} + \frac{Mf_z^2}{2\,EI_z} \right) R\, d\varphi$$

Les déplacements du nœud 2 sont donnés par le théorème de Castigliano :

$$u_2 = \frac{\partial E_{def}}{\partial F_x} \quad , \quad v_2 = \frac{\partial E_{def}}{\partial F_y} \quad , \quad \theta_{z2} = \frac{\partial E_{def}}{\partial M_z}$$

Programme Maple :

```
# étude d'un arc plan

restart:with(linalg): # initialisation

angle_arc:=3*Pi/2;

# efforts dans la section phi

c:=cos(phi):s:=sin(phi):
N:=Fx*s-Fy*c;
T:=Fx*c+Fy*s;
Mfz:=(Fx*s+Fy*(1-c))*R+Mz;

# énergie de déformation

EN:=int(N^2/(2*E*A)*R,phi=0..angle_arc);
ET:=int(T^2/(2*G*k*A)*R,phi=0..angle_arc);
EMfz:=int(Mfz^2/(2*E*Iz)*R,phi=0..angle_arc);
Edef:=EN+ET+EMfz;

# déplacements du noeud 2

u2:=diff(Edef,Fx);
v2:=diff(Edef,Fy);
rotz2:=diff(Edef,Mz);
```

6 Étude d'un arc plan : influence de l'effort normal et de l'effort tranchant

6.1 Énoncé

L'arc $(1-2)$ de centre O et de rayon moyen R est encastré en 1. Il a une section droite constante. Soient A , I_z et k les caractéristiques de la section droite. E et ν sont respectivement le module de Young et le coefficient de Poisson du matériau.



L'arc est soumis en 2 à une force horizontale de composantes $(F_x, 0, 0)$.

1. Calculer le déplacement horizontal du nœud 2.
2. La section droite est rectangulaire :

$$A = b\,h \quad , \quad I_z = \frac{b\,h^3}{12} \quad , \quad k = \frac{10\,(1+\nu)}{12+11\,\nu}$$

   Étudier l'influence de l'effort tranchant en fonction de $\dfrac{R}{h}$ pour plusieurs valeurs du coefficient de Poisson.
3. Étudier l'influence de l'effort tranchant et de l'effort normal en fonction de $\dfrac{R}{h}$ pour $\nu = 0.25$

6.2 Solution

Les efforts dans la section φ sont :

$$N(\varphi) = F_x \sin\varphi \quad , \quad T(\varphi) = F_x \cos\varphi \quad , \quad Mf_z(\varphi) = F_x\, R \sin\varphi$$

L'énergie de déformation est égale à :

$$E_{def} = \int_0^{3\,\pi/2} dE_{def} \quad \text{avec} \quad dE_{def} = \left(\frac{N^2}{2\,EA} + \frac{T^2}{2\,GA\,k} + \frac{Mf_z^2}{2\,EI_z} \right) R\, d\varphi$$

Le déplacement horizontal du nœud 2 est donné par le théorème de Castigliano :

$$u_2 = \frac{\partial E_{def}}{\partial F_x}$$

Programme Maple :

```
# étude d'un arc plan

restart: # initialisation

# angle de l'arc

angle_arc:=3*Pi/2;

# efforts dans la section phi

N:=F*sin(phi); T:=F*cos(phi); Mfz:=F*R*sin(phi);

# énergie de déformation

EN:=int(N^2/(2*E*A)*R,phi=0..angle_arc);
ET:=int(T^2/(2*G*k*A)*R,phi=0..angle_arc);
EMfz:=int(Mfz^2/(2*E*Iz)*R,phi=0..angle_arc);
Edef:=EN+ET+EMfz;

# déplacement

vN:=diff(EN,F); vT:=diff(ET,F); vMfz:=diff(EMfz,F);
v:=vN+vT+vMfz;

# influence de l'effort normal et de l'effort tranchant

G:=E/2/(1+nu);
k:=10*(1+nu)/(12+11*nu); A:=b*h: Iz:=b*h^3/12:

R:=x*h:
fnor:=simplify(vN/v); ftra:=simplify(vT/v);
f1:=subs(nu=0,ftra);f2:=subs(nu=0.25,ftra);f3:=subs(nu=0.5,ftra);
plot([f1*100,f2*100,f3*100],x=2..10,
legend=["nu=0","nu=0.25","nu=0.5"],
title="Influence de l'effort tranchant en % en fonction du coefficient de Poisson",
labels=["R / h","vT/v en %"],
color=[black,red,green],thickness=2);

nu:=0.25;

plot([fnor*100,ftra*100],x=2..10,
legend=["effort normal","effort tranchant"],
title="Influence de l'effort normal et de l'effort tranchant en % (nu=0.25)",
labels=["R / h",""],
thickness=2);
```